Demo 2 – Diode Circuit Simulation in LTspice

XMUT204 Electronic Design

Overview

1. Manufacturer datasheet of a diode.
2. Default diode model in Spice.
3. Customised diode model in Spice.
4. Comparing diode model in LTspice.
5. Improving further the diode model.
6. Modelling other types of diodes.

Example diode

- 1N4000 series diode family (1N4001-1N4007) i.e. 1N4004 diode.

- General-purpose applications in electronics e.g. small power low voltage circuit application, signal rectification, etc.



Partial Datasheet of 1N4004

- Some information have been extracted from a 1N4004 data sheet as shown in the figure given below.



Max avg rectified current I_O (A)	1	Forward voltage drop V_F (V)	1
Peak repetitive reverse voltage V_{RRM} (V)	400	@ I_F (A)	1
Peak forward surge current I_{FSM} (A)	30	Max reverse current I_R (µA)	5
Total capacitance C_T (pF)	15	@ V_R (V)	400

Modelling Diode in SPICE

- Using default model of diode in SPICE - most of parameters associated with a generic diode’s characteristics and their relevant default values.
- If you require specific diode model, you can choose any of the following:
  - Consult the manufacturer’s web site for model and specifications.
  - Build a SPICE model from those parameters listed on the manufacturer data sheet.
  - Take measurements of an actual device. Then, calculate, compare and adjust the SPICE parameters to the measurements.

Modelling Diode in SPICE (cont.)

- The diode statement begins with a diode element name which must begin with "`d`" plus optional characters.
- Example diode element names include: `d1`, `d2`, `dtest`, `da`, `db`, `d101`.
- Two node numbers specify the connection of the anode and cathode, respectively, to other components.
- General form:

```
d[name] [anode] [cathode] [modelname]

.model ([modelname] d [parmtr1=x] [parmtr2=y] . . .)
```

Modelling Diode in SPICE (cont.)

- The node numbers are followed by a model name, referring to a subsequent "`.model`" statement.
- The model statement line begins with "`.model`", followed by the model name matching one or more diode statements.
- Next, a "`d`" indicates a diode is being modelled.
- The remainder of the model statement is a list of optional diode parameters of the form:

```
ParameterName=ParameterValue
```

Example 1:

```
d1 1 2 mod1
.model mod1 d
```

Example 2:

```
D2 1 2 Da1N4004
.model Da1N4004 D (IS=18.8n RS=0 BV=400 IBV=5.00u CJO=30
+M=0.333 N=2)
```

Modelling Diode in SPICE (cont.)

- Default diode SPICE parameters

Symbol	Name	Parameter	Units	Default
IS	IS	Saturation current (diode equation)	A	1E-14
RS	RS	Parasitic resistance (series resistance)	Ω	0
n	N	Emission coefficient, 1 to 2	-	1
τD	TT	Transit time	s	0
CD(0)	CJO	Zero-bias junction capacitance	F	0
ϕ0	VJ	Junction potential	V	1
m	M	Junction grading coefficient:	-	0.5
		• 0.33 for linearly graded junction	-	-
		• 0.5 for abrupt junction	-	-
Eg	EG	Activation energy:	eV	1.11
		• Si: 1.11	-	-
		• Ge: 0.67	-	-
		• Schottky: 0.69	-	-
pi	XTI	IS temperature exponent:	-	3.0
		• pn junction: 3.0	-	-
		• Schottky: 2.0	-	-
kf	KF	Flicker noise coefficient	-	0
af	AF	Flicker noise exponent	-	1
FC	FC	Forward bias depletion capacitance coefficient	-	0.5
BV	BV	Reverse breakdown voltage	V	∞
IBV	IBV	Reverse breakdown current	A	1E-3

Modelling Diode in SPICE (cont.)

- SPICE parameters for selected diodes (note: sk=Schottky; Ge=germanium; else silicon).

Part	IS	RS	N	TT	CJO	M	VJ	EG	XTI	BV	IBV
Default	1E-14	0	1	0	0	0.5	1	1.11	3	∞	1m
1N5711 sk	315n	2.8	2.03	1.44n	2.00p	0.333	-	0.69	2	70	10u
1N5712 sk	680p	12	1.003	50p	1.0p	0.5	0.6	0.69	2	20	-
1N34 Ge	200p	84m	2.19	144n	4.82p	0.333	0.75	0.67	-	60	15u
1N4148	35p	64m	1.24	5.0n	4.0p	0.285	0.6	-	-	75	-
1N3891	63n	9.6m	2	110n	114p	0.255	0.6	-	-	250	-
10A04 10A	844n	2.06m	2.06	4.32u	277p	0.333	-	-	-	400	10u
1N4004 1A	76.9n	42.2m	1.45	4.32u	39.8p	0.333	-	-	-	400	5u
1N4004 data sheet	18.8n	-	2	-	30p	0.333	-	-	-	400	5u

Deriving the SPICE Models from Specification Sheets

a. Emission Coefficient Parameter (N)
b. Saturation Current (IS)
c. Parasitic Resistance of Diode (RS)
d. Transit Time (TT)
e. Junction Capacitance (CJO)
f. Junction Grading Coefficient (M)
g. Junction Potential (VJ)
h. Activation Energy (EG)
i. Temperature Exponent (XTI)
j. Reverse Breakdown Current (IBV)
k. Breakdown Voltage (BV)

Deriving the SPICE Models from Specification Sheets (N parameter)

- The emission coefficient or ideality factor n accounts for the effect of recombination of holes with free electrons in the depletion region of the diode.
- Select a value for SPICE emission coefficient of the diode parameter N between 1 and 2.
- The value is used to estimate n in the Eber-Molls equation of a diode:

$$ID = IS(e^{VD/nVT} - 1)$$

Deriving the SPICE Models from Specification Sheets (VD and IS parameters)

- The saturation current, IS, is derived from the diode equation, a value of (VD, ID) on the graph as given in the Figure 1 before.
- Then, N = 2 (i.e. n in the diode equation) – referring to the Eber-Molls equation given below.

$$ID = IS(e^{VD/nVT} - 1)$$

- From the theory, VT = 26 mV (at 25°C) and n = 2.0. Furthermore, we obtain that VD = 0.925 V at 1 A from graph in Figure 1.



- As a result, by entering all of these values, the saturation current of the diode is found to be:

$$1\ A = IS(e^{(0.925\ V)/(2)(26\ mV)} - 1)$$
$$IS = 18.8E-9$$

- The numerical values of IS = 18.8 nA and N = 2

Deriving the SPICE Models from Specification Sheets (RS parameter)

- The parasitic series resistance, RS default is assigned to 0 Ω for now, and it will be estimated later.
- That resistance represents the behavior of the interface between the diode semiconductor material and the metal contact used for the terminal of the diode.
- Unlike resistors, though, diodes are not linear devices. This means that the resistance of diodes does not vary directly and proportional to the amount of voltage and current applied to them.
- An actual diode offers a very small resistance (not zero) when forward biased and is called a forward resistance
- Some other parasitic resistance present because the impedance due to the parasitic capacitance and inductance in the diode.

Deriving the SPICE Models from Specification Sheets (TT parameter)

- When a forward-biased diode has a reverse voltage applied across it, it takes time for the charge to dissipate and hence for the diode to turn off. The time taken for the diode to turn off is captured primarily by the transit time parameter. Once the diode is off, any remaining charge then dissipates, the rate at which this happens being determined by the carrier lifetime.
- It is suggested that the transit time, TT, τD, be approximated from the reverse recovery stored charge, QRR, a data sheet parameter (i.e. not available in our data sheet) and IF, forward current.

$$ID = IS(e^{VD/nVT} - 1) \quad \tau D = QRR/IF$$

- We take the TT = 0 sec, default for lack of QRR.
- Though it would be reasonable to take TT for a similar rectifier like the 10A04 at 4.32 us.
- In the table of several diode models, the 1N3891's TT is not a valid choice because it is a fast recovery rectifier.

Deriving the SPICE Models from Specification Sheets (CJO parameter)

- This parameter is about the parasitic capacitance that exist in the diode.
- The zero bias junction capacitance, CJO, is estimated from the VR vs CJ graph in graph from datasheet.
- In the graph, the capacitance at the nearest to zero voltage on the graph is 30 pF at 1 V.



- If simulating high-speed transient response, as in switching regulator power supplies, TT and CJO parameters must be provided.

Deriving the SPICE Models from Specification Sheets (M parameter)

- The junction grading coefficient, M is related to the doping profile of the junction i.e. this is not a datasheet item.
- The default is 0.5 for an abrupt junction.
- We opt for M = 0.333 corresponding to a linearly graded junction.
- The power rectifiers in the table given above use lower values for M than 0.5.

Deriving the SPICE Models from Specification Sheets (VJ & EG parameters)

- Junction potential is the voltage drop across the depletion region of the PN area of the diode.
- EG is the energy gap for the semiconductor type measured in Joules (J). The value for silicon is usually taken to be 1.11 eV, where 1 eV is 1.602e-19.
- We take the default values for junction potential (VJ) and activation energy (EG). Many diodes use VJ = 0.6.
- However, the 10A04 rectifier uses the default, which we use for our 1N4004 model (Da1N4001).
- Use the default activation energy, EG = 1.11 for silicon diodes and rectifiers (The given table also lists values for Schottky and germanium diodes).

Deriving the SPICE Models from Specification Sheets (XTI parameter)

- XTI is the saturation current temperature exponent. This is usually set to 3.0 for pn-junction diodes, and 2.0 for Schottky barrier diodes.
- Appropriate values for XTI and EG depend on the type of diode and the semiconductor material used.
- In practice, the values of XTI and EG need tuning to model the exact behaviour of a particular diode.
- Take the saturation current temperature exponent, XTI = 3, the default IS temperature coefficient for silicon devices.
- See the table of several diode models for XTI for Schottky diodes.

Deriving the SPICE Models from Specification Sheets (IBV and BV parameters)

- These parameters refer to the current (IBV) and voltage (BV) whenever diode is operated in the reverse bias mode.
- In the abbreviated data sheet, as shown in the figure above, lists IR = 5 µA @ VR = 400 V, corresponding to IBV = 5 uA and BV = 400 V respectively.



Max avg rectified current I_O (A)	1	Forward voltage drop V_F (V)	1
Peak repetitive reverse voltage V_{RRM} (V)	400	@ I_F (A)	1
Peak forward surge current I_{FSM} (A)	30	Max reverse current I_R (µA)	5
Total capacitance C_T (pF)	15	@ V_R (V)	400

Deriving the SPICE Models from Specification Sheets (IBV and BV parameters)

- BV is only necessary if the simulation exceeds the reverse breakdown voltage of the diode, as is the case for zener diodes.
- IBV, reverse breakdown current, is frequently omitted, but may be entered if provided with BV.



Comparing Diode Models from Different Sources (First Trial)

- Figure below shows a circuit to compare the manufacturers model, the model derived from the datasheet, and the default model using default parameters.
- SPICE circuit for comparison of manufacturer model (D1), calculated datasheet model (D2), and default model (D3).



- The three dummy 0 V sources are necessary for diode current measurement.
- The 1 V source is swept from 0 to 1.4 V in 0.2 mV steps (e.g. see `.DC` statement in the netlist in Table below). DI1N4004 is the manufacturer's diode model, Da1N4004 is our derived diode model.

Comparing Diode Models from Different Sources (First Trial)

- SPICE netlist parameters of three diode SPICE models: (D1) DI1N4004 manufacturer's model, (D2) Da1N40004 datasheet derived, (D3) default diode model.

```
.DC V1 0 1400mV 0.2m

.model Da1N4004 D (IS=18.8n RS=0 BV=400 IBV=5.00u CJO=30
+M=0.333 N=2.0 TT=0)

.model DI1N4004 D (IS=76.9n RS=42.0m BV=400 IBV=5.00u CJO=39.8p
+M=0.333 N=1.45 TT=4.32u)

.model Default D
```

Comparing Diode Models from Different Sources (First Trial)

- We compare the three models in Figure below for the forward voltage characteristics of the diode as in the datasheet graph data in Table below.
- More specifically, VD is the diode voltage versus the diode currents
- We compare the characteristics for the models of the manufacturer's model, our calculated datasheet model and the default diode model.



Comparing Diode Models from Different Sources (First Trial)

- Plots of the first simulation results (manufacturer model – top, model from datasheet – middle, and default diode model – bottom).







Comparing Diode Models from Different Sources (First Trial)

#	VD	Manufacturer model	Datasheet model	Default model	1N4004 graph
1	0.700	1.612924	0.01416211	0.005674683	0.01
2	0.800	3.346832	0.09825960	0.2731709	0.13
3	0.900	5.310740	0.6764928	12.94824	0.70
4	0.925	5.823654	1.096870	34.04037	1.00
5	1.000	7.395953	4.675526	618.5078	2.00
6	1.100	9.548779	30.231452	2,9544.71	3.30
7	1.200	11.74489	223.3392	1,411,283	5.30
8	1.300	13.97087	1,543.591	67,413,790	8.00
9	1.400	16.21861	10,668.40	3,220,203,000	12.0

- The last column “1N4004 graph” is from the datasheet voltage versus current curve in manufacturer datasheet which we attempt to match.
- Comparison of the currents-voltage curve for the three models shows that the default model is good at low currents, the manufacturer’s model is good at high currents, and our calculated datasheet model is best of all up to 1 A.
- Agreement is almost perfect at 1 A because the IS calculation is based on diode voltage at 1 A.
- But in the end, our model grossly over states current above 1 A as shown in Figure 3 shown above.

Comparing Diode Models from Different Sources (Second Trial)

- Second trial is to improve calculated datasheet model compared with manufacturer model and default model. The solution is to increase dynamic resistance of the diode, RS from the default RS = 0 Ω.
- Changing RS from 0 Ω to 8 mΩ in the datasheet model causes the curve to intersect 10 A (not shown) at the same voltage as the manufacturer's model.
- Increasing RS to 28.6 mΩ shifts the curve further to the right. This has the effect of more closely matching our datasheet model to the datasheet graph.





Comparing Diode Models from Different Sources (Second Trial)

- Changing Da1N4004 model statement RS = 0 Ω to RS = 28.6 mΩ decreases the current at VD = 1.4 V to 12.2 A.

```
.model Da1N4004 D (IS=18.8n RS=28.6m BV=400 IBV=5.00u CJO=30
+M=0.333 N=2.0 TT=0)
```

- Referring to the lecture of diode characteristics and pn junction, adding the ideal + voltage drop model with dynamic resistance of the diode will improve the model further.
- The bulk of chunk of dynamic resistance is due to series resistance when diode is forward biased.
- Other parasitic resistance can be added to improve the model, but these will not increase the accuracy of the model significantly.

Comparing Diode Models from Different Sources (Second Trial)

Plots of the second simulation results (manufacturer model – top, model from datasheet – middle, and default diode model – bottom).







Comparing Diode Models from Different Sources (Second Trial)

- Table below shows that the current 12.2447 A at 1.4 V matches the graph at 12 A.
- However, the current at 0.925 V has degraded from 1.096870 A above to 0.7318536 A.

#	VD	Manufacturer model	Datasheet model	1N4001 graph
1	0.701	1.628276	0.01432463	0.01
2	0.800	3.343072	0.09297594	0.13
3	0.900	5.310740	0.5102139	0.70
4	0.925	5.823654	0.7318536	1.00
5	1.000	7.395953	1.763520	2.00
6	1.100	9.548779	3.848553	3.30
7	1.200	11.74489	6.419621	5.30
8	1.300	13.97087	9.254581	8.00
9	1.400	16.21861	12.24470	12.00

Modelling Zener Diode

There are two approaches to modeling a zener diode:

- set the BV parameter to the zener voltage in the model statement, or
- model the zener with a subcircuit containing a diode clamper set to the zener voltage.

```
.model D1N4469 D (
BV=15 IBV=17m )
```

```
.subckt DI-1N4744A 1 2
* Terminals A K
D1 1 2 DF
DZ 3 1 DR
VZ 2 3 13.7
.mode DF D ( IS=27.5p RS=0.620 N=1.10
+ CJO=78.3p VJ=1.00 M=0.330 TT=50.1n )
.model DR D ( IS=5.49f RS=0.804 N=1.77 )
.end
```

Modelling Zener Diode (cont.)

- An example of the first approach sets the breakdown voltage BV to 15 V for the 1n4469 15 V zener diode model (IBV optional):

```
.model D1N4469 D ( BV=15 IBV=17m )
```





Modelling Zener Diode (cont.)

- The second approach models the zener with a subcircuit. Clamper D1 and VZ in
  Figure below models the 15 V reverse breakdown voltage of a 1N4477A zener diode. Diode DR accounts for the forward conduction of the zener in the subcircuit.



- Zener diode subcircuit uses clamper (D1 and VZ) to model zener.

```
.subckt DI-1N4744A 1 2
* Terminals A K
D1 1 2 DF
DZ 3 1 DR
VZ 2 3 13.7
.model DF D ( IS=27.5p RS=0.620 N=1.10
+ CJO=78.3p VJ=1.00 M=0.330 TT=50.1n )
.model DR D ( IS=5.49f RS=0.804 N=1.77 )
.END
```



Modelling Zener Diode (cont.)

- For the zener diode given above, the following shows its SPICE directive.

```
.SUBCKT DI-1N4744A 1 2
* Terminals A K
D1 1 2 DF
DZ 3 1 DR
VZ 2 3 13.7
.MODEL DF D ( IS=27.5p RS=0.620 N=1.10
+ CJO=78.3p VJ=1.00 M=0.330 TT=50.1n )
.MODEL DR D ( IS=5.49f RS=0.804 N=1.77 )
.END
```



Modelling Other Types of Diode

You might see also that there are other types of diodes in the market. Model of these other diodes require different modelling in the SPICE.

- *Tunnel diode*: A tunnel diode may be modeled by a pair of field effect transistors (JFET) in a SPICE subcircuit. This diode is typically used in the oscillator circuits.
- *Gunn diode*: A Gunn diode may also be modeled by a pair of JFET's. This diode is used for microwave relaxation oscillator applications.

Review Summary

a. Diodes are described in SPICE by a diode component statement referring to `.model` statement. The `.model` statement contains parameters describing the diode. If parameters are not provided, the model takes on default values.

b. Static DC parameters include N, IS, and RS. Reverse breakdown parameters: BV, IBV.

c. Accurate dynamic timing requires TT and CJO parameters.

d. Models provided by the manufacturer are highly recommended.